

Construire un cadran solaire horizontal

Ce cadran, calculé pour une latitude précise, corrigé de l'équation du temps et de l'écart au méridien, donne l'heure solaire juste

Cet article a été publié dans les n° 23 et 24 de la revue Astro Ciel en 1989

Michel Derouet et Daniel Deneuchatel

Le déclic fut la lecture d'un article de Jearl WALKER dans « Pour la Science » de février 1981. Celui-ci relatait la mise au point d'un « cadran solaire analemmatique » par un Américain, C. SLOAN, à Arapahoe en Caroline du Nord. Après en avoir exposé succinctement le principe, Jearl WALKER terminait son article en fournissant un mini-programme en BASIC pour calculer la longueur des ombres portées.

Il était intéressant de décortiquer et comprendre chaque composante des équations. Ô surprise !... une des premières, victime d'une malencontreuse coquille, était fautive ! Il n'en fallait pas plus pour exciter la curiosité.

Tout d'abord, finissons-en avec la terminologie. On trouve quelquefois l'adjectif « analemmatique ». Les difficultés que j'ai parfois rencontrées dans les calculs n'étaient pas « problématiques », mais bien problématiques ! Robert SAGOT, dans l'Astronomie d'octobre 1983 nous rappelle que depuis VAUZELARD (1640), on appelle cadran analemmatique tout

cadran pour lequel la lecture de l'heure s'effectue à l'intersection d'un style rectiligne et mobile avec une ellipse graduée. Il en existe de nombreuses variétés, et celui que nous proposons de réaliser, bien que d'inspiration et de conception totalement différentes, possède la même finalité.

Le dessin de ce type de cadran horizontal corrige les effets de la latitude, de la longitude, de la déclinaison du Soleil, et permet ainsi de lire l'heure solaire et l'heure légale d'un lieu tous les jours de l'année.

RAPPEL DE NOTIONS COSMOLOGIQUES

Pour une information plus complète, lire l'article de Gérard OUDENOT dans l'Astronomie d'octobre 1983.

Du fait de la rotation de la Terre sur son axe, l'ensemble du ciel accomplit un tour en 23 h 56 m 04 s... C'est le jour sidéral. La Terre se déplace sur son orbite autour du Soleil en une année. Ceci a pour conséquence que, pendant

qu'elle accomplit un tour sur elle-même, elle a parcouru environ 2,5 millions de kilomètres sur son orbite. Il faut donc qu'elle tourne un peu plus sur elle-même pour revenir en face du Soleil. C'est le jour solaire que nous connaissons et qui vaut environ 24 h.

L'axe de la Terre est incliné de $23^{\circ}27'$ par rapport à la perpendiculaire du plan de son orbite autour du Soleil. Ceci entraîne une hauteur variable du Soleil au cours de l'année et une inégalité des jours et des nuits. La première conséquence de tous ces mouvements est que la longueur de l'ombre portée d'un objet vertical, sur un plan horizontal, varie au cours d'une année. Elle sera maximum au solstice d'hiver et minimum au solstice d'été.

ÉQUATION DU TEMPS

C'est grâce aux observations très précises de Tycho BRAHE que Johannes KEPLER au début du XVII^e siècle put mettre en évidence les lois qui portent son nom :

- la Terre se déplace sur une ellipse




FIGURE 1. Équation du temps (en minutes). D'après les Ephémérides du Bureau des Longitudes.

dont le Soleil occupe un des foyers.
- le rayon vecteur Terre-Soleil balaie des aires égales en des temps égaux.

La Terre se déplace donc sur son orbite à une vitesse variable. C'est-à-dire que le jour vrai (intervalle de temps qui sépare deux passages consécutifs du Soleil au méridien) varie de 23 h 59 m 39 s à 24 h 00 m 30 s. De plus, l'inclinaison de l'écliptique par rapport à l'équateur fait que le déplacement

analemme (la définition de l'analemmes semble évolutive au cours des siècles). Ils seront intégrés dans le dessin pour chaque heure de la journée.

LONGITUDE

Le midi d'un cadran se lit sur la méridienne de temps moyen se rapportant au midi solaire. Or, pour que le midi moyen corresponde au midi légal, il faut tenir compte du décalage du lieu en longitude par rapport au méridien de référence... Greenwich en ce qui nous concerne. La méridienne sera donc déportée à l'Est du midi si le lieu d'implantation est à l'Est... et inversement.

ARCS DIURNES

Le Soleil se déplace dans le ciel en décrivant chaque jour un arc de cercle sur la sphère céleste. L'ombre d'un point va donc décrire un cône au cours de la journée. L'intersection de ce cône et de la surface horizontale du cadran va générer une conique. Sous nos latitudes, ce sera un arc d'hyperbole. Le Soleil se déplaçant dans le ciel à une hauteur différente selon les saisons, on pourra donc tracer plusieurs arcs diurnes qui, entre autres, permettront selon la longueur de l'ombre de donner une idée approximative de la date. On tracera :

- Les arcs des deux solstices qui seront les limites Nord et Sud de l'ombre.
- L'arc des équinoxes, confondus en une seule droite car le Soleil possède alors la même déclinaison. Il se trouve dans le plan perpendiculaire à l'axe terrestre (déclinaison = zéro) et l'ombre suit une ligne droite.
- Dix à douze autres arcs dont on choisira judicieusement les dates espacées de 10 à 20 jours de façon à obtenir un dessin clair, lisible et esthétique. Chacun de ces arcs comportera deux dates : celles où le Soleil a, au cours de l'année, la même déclinaison.

CALCULS ET DESSIN DU CADRAN

Pour faciliter la compréhension du raisonnement et des formules présentées plus loin, il faut savoir que tout a été ramené à la hauteur du Soleil au-dessus de l'horizon, quelles que soient l'heure et la saison. Le générateur de l'ombre est un style fixe, rectiligne et vertical dont la hauteur variera en fonction de la taille désirée du cadran. Ce que nous appelons style est en fait, gnomoniquement parlant, la sous-style. Avec un style de 50 mm, le dessin s'inscrit dans un rectangle de 60 cm x 35 cm. Il faudra de toutes façons limiter la taille du cadran car lorsque le Soleil est bas sur l'horizon, les ombres deviennent vite démesurées. Les abréviations employées plus loin ne sont pas très orthodoxes... Elles ont été choisies pour leurs propriétés mnémotechniques.

COORDONNÉES DU LIEU

Sur une carte de l'I.G.N. au 1/25 000, repérer le lieu d'implantation du cadran. Ensuite, à l'aide des échelles de pourtour de carte (référées au méridien international), déterminer avec précision (degrés, minutes et secondes ces deux dernières seront transformées en valeur décimale) les coordonnées en latitude et longitude.

POINT D'ECLIPTIQUE

Tracer d'abord la ligne de midi. C'est une droite orientée Nord-Sud. Ensuite, sur celle-ci, situer le premier repère qui sera l'emplacement du style. Enfin, déterminer le deuxième repère situé au Sud du style en multipliant la hauteur de celui-ci par la cotangente de la latitude du lieu. On obtient ainsi le point d'écliptique. C'est à partir de ce point que doivent être tracées les lignes radiales des heures. Ce point ne sert qu'au traçage, il ne figurera donc pas dans le dessin final.

DROITE D'EQUINOXE

Les jours d'équinoxe, le Soleil est




FIGURE 2. Méridienne de temps moyen. (Ici pour un cadran vertical). D'après « Midi au Soleil » de Jean FULCRAND et Pierre BOURGE. apparent du Soleil est plus rapide aux solstices qu'aux équinoxes.

L'addition de ces deux anomalies constitue un écart entre le temps solaire vrai et le temps solaire moyen. Cet écart est l'équation du temps.

On voit qu'à certaines périodes de l'année, cet écart peut atteindre près de 15 minutes ; il n'est donc pas possible de le négliger. Ceci va nous conduire à introduire une correction pour que, quelles que soient les saisons, l'heure du cadran coïncide avec l'heure moyenne. Sur le dessin du cadran, cette correction va prendre la forme d'un huit allongé et dissymétrique qui s'appelle la méridienne de temps moyen ou




FIGURE 3. Equinoxes et droites horaires.

dans le plan de l'équateur terrestre. Sa hauteur dans le ciel est égale à 90° moins la latitude du lieu. La pointe de l'ombre suit alors une ligne perpendiculaire à la ligne des midis, située au Nord du style à une distance de celui-ci égale à :

Hauteur du style x tangente de la latitude.

DROITE DES HEURES

Elles se tracent à partir du point d'écliptique et passent par un autre point situé sur la droite d'équinoxe à une certaine distance de la droite de midi :

Distance au midi sur la ligne d'équinoxe

ST

$\text{tg } H \times \text{-----}$

Cos LA

où H : heure (13 h = 15° , 14 h = 30° , etc...)

ST : hauteur du style

LA : latitude

Il est préférable d'en tracer une toute les demi-heures (de 6 h 30 à 17 h 30), celles du matin sont symétriques à celles de l'après-midi par rapport au midi. Celles de 6 h et 18 h sont perpendiculaires au midi et passent par le point d'écliptique.

ARCS DIURNES

Ce sont les hyperboles des dates. Il faut sélectionner les dates. Ce choix est laissé à la fantaisie de chacun. On peut privilégier les anniversaires, les fêtes nationales ou religieuses, etc... Pour cela, utiliser les tables du Bureau des

Longitudes. Voici, à titre d'exemple, un tableau avec 26 dates (14 arcs diurnes), la déclinaison du Soleil et l'équation du temps en degré (1 degré = 4 minutes). Cette dernière servira plus tard pour le calcul des analemmes.

D : déclinaison du Soleil

H : heure en degré (1 heure = 15°)

Attention : Sin zéro = 0 ; Cos zéro =

Pour plus de détails, voir le chapitre

Dates	Déclinaison	Eq. temps 1 ^{re} date	Eq. temps 2 ^e date
21/12	- 23,45	- 0,50	-
10/01 - 03/12	- 22,06	+ 1,90	- 2,50
21/01 - 22/11	- 20,06	+ 2,75	- 3,50
01/02 - 11/11	- 17,25	+ 3,40	- 4,00
10/02 - 02/11	- 14,55	+ 3,60	- 4,10
21/02 - 22/10	- 10,80	+ 3,45	- 3,95
01/03 - 14/10	- 7,85	+ 3,12	- 3,50
11/03 - 04/10	- 4,00	+ 2,50	- 2,75
21/03 - 23/09	0	+ 1,80	- 1,75
01/04 - 13/09	+ 4,10	+ 1,00	- 1,00
16/04 - 29/08	+ 9,50	0	+ 0,25
01/05 - 13/08	+ 14,85	- 0,70	+ 1,25
15/05 - 30/07	+ 18,70	- 1,20	+ 1,60
21/06	+ 23,45	+ 40	-

Le calcul se fait en deux temps :

- D'abord, calculer la hauteur du Soleil au-dessus de l'horizon à chaque demi-heure pour chacune des dates retenues. Pour cela, on passe des coordonnées équatoriales (déclinaison) aux coordonnées horizontales par la formule suivante :

$$\sin HS = \sin LA \cdot \sin D + \cos LA \cdot \cos D \cdot \cos H$$

où HS : hauteur du Soleil sur l'horizon
LA : latitude du lieu

« changement de coordonnées » dans les Ephémérides du Bureau des Longitudes. Faire le calcul pour chaque demi-heure de 12 h à 18 h...; pour le matin, les valeurs sont symétriques.

- Ensuite, calculer la longueur de l'ombre :

$$M = \frac{ST}{\text{tg } HS}$$

où M : longueur de l'ombre
ST : hauteur du style
HS : hauteur du Soleil




FIGURE 4. Arcs diurnes.

Reporter les longueurs de l'ombre obtenues pour chaque date sur les droites des demi-heures à l'aide d'un compas centré sur l'emplacement du style. En joignant les points d'une même date, on obtient une portion d'hyperbole... en hiver, incurvée vers le Nord et en été vers le Sud.

A ce stade, notre cadran permet de lire l'heure solaire. Mais notre but était de lire aussi l'heure légale.

CALCUL ET DESSIN DES ANALEMMES

Nous allons donc introduire et adapter les analemmes à chaque heure du cadran de 7 h à 17 h. Pour l'étape qui va suivre, il serait souhaitable d'abandonner la calculatrice pour avoir recours à un micro-ordinateur disposant d'un logiciel tableur, genre Multiplan ou autres. Pour ceux que les calculs rebutteraient, il est possible d'utiliser la technique du traçage de la méridienne décrite dans « Midi au Soleil ». Les calculs ne sont pas plus complexes que les précédents mais plus longs.

Pour ce faire, il faut reprendre les trois opérations qui ont conduit au calcul de la longueur de l'ombre pour les arcs diurnes :

- Nouvel angle horaire,
- Hauteur du Soleil sur l'horizon,
- Longueur de l'ombre à partir du style.

Nous pourrions nous en tenir à cette dernière valeur et la reporter à nouveau à l'aide d'un compas ; mais pour les heures proches de midi le trait de compas serait trop tangent aux arcs diurnes et perdrait beaucoup en précision. Ceci nous amène à calculer d'autres paramètres et à procéder au traçage en deux temps :

- à l'aide du compas centré sur le style, nous tracerons un premier repère sur la ligne de midi. (Voir calcul distance au style)
- à partir de ce repère, nous en tracerons un second (voir calcul distance au midi) qui sera perpendiculaire à la ligne des midis et ira couper l'arc diurne considéré à l'endroit désiré.

La figure ci-contre permet de comprendre les calculs intermédiaires et la façon de procéder.

NOUVEL ANGLE HORAIRE

Avant d'introduire l'écart en longitude, il faut d'abord définir la valeur algébrique (+ ou -) à lui accorder selon que l'on est à l'Est ou à l'Ouest du méridien de référence. Nous transformerons partiellement et adapterons à nos besoins la recommandation de l'Union Astronomique Internationale (1982). Lorsque le cadran est situé à l'Est du méridien, l'écart en longitude sera positif et à l'Ouest négatif. Pour un pays autre que la France, prendre comme référence le méridien sur lequel est centré le fuseau horaire.




FIGURE 5. Exemple pour le calcul de l'ombre, le 11 novembre, à 14 h 00.

La valeur de l'équation du temps est à prendre également en valeur algébrique. Cette fois, il n'y a plus symétrie entre les heures du matin et du

pour le calcul des arcs diurnes :
 $\sin HS = \sin LA \cdot \sin D + \cos LA \cdot \cos D \cdot \cos H2$

07 h à 11 h	- H + LO - EQ	} = H2 en degrés
13 h à 17 h	H + LO - EQ	
à midi (H = zéro)	+ LO - EQ	
où H = heure en degrés	} en degré (valeur algébrique)	
LO = écart en longitude		
EQ = équation du temps		

soir. Il y aura donc deux formules... et même trois !

HAUTEUR DU SOLEIL DANS LE CIEL

On reprend la formule déjà utilisée

où LA : Latitude
D : déclinaison du Soleil
H2 : nouvel angle horaire
HS : hauteur du Soleil

LONGUEUR DE L'OMBRE

Mesurée à partir de la base du style.

$$\text{Longueur de l'ombre} = \frac{\text{ST}}{\text{tg HS}}$$

où ST : hauteur du style
HS : hauteur du Soleil

ANGLE A1

C'est l'angle entre la ligne du midi et la droite qui relie le point d'écliptique à la pointe de l'ombre :

$$\text{Tg A1} = \frac{\text{ST}}{\text{tg H2} \times \text{Cos LA}} \quad (\text{ST.tg LA}) + (\text{ST.Ctg LA})$$

Toutes les abréviations sont déjà connues.

ANGLE A2

C'est l'angle entre les droites :
écliptique → pointe de l'ombre et
style → pointe de l'ombre :

$$\text{Sin A2} = \frac{\text{Sin A1} \times (\text{ST.Ctg LA})}{\text{ST}}$$

en simplifiant :

$$\text{Sin A2} = \text{Sin A1} \times \text{Ctg LA} \times \text{Tg HS}$$

ANGLE A3

C'est l'angle entre la ligne des midis et la droite style → pointe de l'ombre. Celui-ci sera négatif pour tous les points qui seront à l'Ouest de midi.

En simplifiant : $A3 = A1 + A2$

Maintenant que l'angle A3 et la longueur de l'ombre sont connus, il est facile de déterminer le point recherché. Cette opération se fait en deux temps.

DISTANCE AU STYLE

A partir du style, tracer un repère sur la ligne des midis à une distance égale à $\text{Cos A3} \times \text{longueur de l'ombre}$.

Cette distance pourra être négative (donc au Sud du style). Ce sera le cas vers le solstice d'été pour 7 h et 17 h.

DISTANCE AU MIDI

Toujours à l'aide du compas, et à partir du premier repère, reporter sur l'arc diurne considéré une longueur égale à : $\text{Sin A3} \times \text{longueur de l'ombre}$. Ce repère sera automatiquement perpendiculaire à la ligne des midis. La distance sera positive à l'Est de midi et négative à l'Ouest.

Par ces deux reports successifs, on obtient une détermination précise de ce point. Pour en vérifier l'emplacement, il est possible de reporter la longueur de l'ombre à partir du style ; le trait du compas viendra recouper le point. Certains résultats vous surprendront peut-être... A certaines heures, la longueur de l'ombre tend vers l'infini,

on dépasse de beaucoup la taille du cadran.

Lorsque, pour une heure donnée, le calcul a été fait pour toutes les dates retenues, il suffit de joindre les intersections pour obtenir une figure en forme de huit. Recommencer pour chaque heure, et on observe que les courbes en 8 se déforment de plus en plus à mesure qu'on s'éloigne du midi. L'axe de celles-ci sera décalé par rapport au trait de l'heure solaire qui s'y rapporte, sauf pour un cadran situé sur le méridien de Greenwich. Pour un cadran situé à plus de 4,12° à l'Ouest et 3,62° à l'Est du méridien de référence, les courbes en 8 seront entièrement à côté de la droite de l'heure solaire. En France, ce sera le cas de la pointe de la Bretagne à l'Ouest de Quimper, et des régions situées à l'Est d'une ligne passant par Valenciennes, Auxerre et Montpellier.

Il ne reste plus qu'à dater les arcs diurnes. Le même arc diurne comporte deux dates (hormis les solstices) et la lecture de l'heure juste s'effectue à l'intersection de la courbe en 8 et de la

Du 1^{er} janvier au 16 avril et du 15 juin au 2 septembre : à l'Ouest.

Du 17 avril au 14 juin et du 3 septembre au 31 décembre : à l'Est.

L'heure se lira donc sur la moitié de la courbe en 8 correspondant au côté de la date. Il s'agit en l'occurrence de l'heure TU !... à laquelle il faut ajouter mentalement une heure en hiver et deux heures en été pour avoir l'heure légale française. On peut également numéroté les heures en heures d'hiver au Nord et heures d'été au Sud. Pour plus de clarté, il est possible de ne faire figurer que les courbes en 8 et les heures correspondantes, ou, si le cadran est très grand, de tracer une courbe en 8 toutes les 30 minutes. Tout ceci est laissé à l'appréciation de chacun.


Quelques exemples sont donnés dans le tableau suivant. Ils ont été calculés pour un style de 50 mm et les coordonnées suivantes :

Latitude : 47°28'01" (47,47°)
Ecart au méridien : + 0°46'06" (+ 0,77°)

	21 juin 17 h	21 janvier 14 h	22 novembre 14 h	10 février 9 h
Déclinaison du Soleil	+ 23,45	- 20,06	- 20,06	- 14,55
Equation du temps	+ 0,40	+ 2,75	- 3,50	+ 3,60
Nouvel angle horaire	75,37	28,02	34,27	- 47,83
Hauteur du Soleil	26,74	17,93	15,78	14,72
Longueur de l'ombre	99,25	154,54	176,88	190,20
Angle A1	70,49	21,41	26,66	- 39,12
Angle A2	25,82	6,21	6,68	- 8,75
Angle A3	96,32	27,63	33,34	- 47,88
Distance au style	- 10,92	136,92	147,77	127,62
Distance au midi	98,65	71,67	97,22	- 141,14

pointe de l'ombre. Il faut donc distribuer les dates conformément à la répartition des mois sur la méridienne de temps moyen :

Il ne reste plus maintenant qu'à le graver dans un matériau noble et durable... ce sera l'occasion d'un prochain article !



« Brouillon » réalisé sur contreplaqué. Il est 13 h 00 TU soit 14 h 00 (heure légale) le 22 octobre.

BIBLIOGRAPHIE

- Midi au Soleil - Jean FULCRAND - Pierre BOURGE.
- Pour la Science - Février 1981.
- Ephémérides Astronomiques 1986 et 1987. Bureau des Longitudes.
- Astronomie. L'Encyclopédie Atlas du Ciel. n° 113 et 114.
- L'Astronomie. Bulletin de la S.A.F. Octobre 1983.

Je souhaiterais que les personnes qui seraient tentées après la lecture de cet article me fassent part de leurs suggestions et réalisations.

DEROUET Michel, La Bouquinière, 37380 MONNAIE.

ERRATUM

Dans la première partie de l'article (Astro-Ciel n° 23), au tableau de la page 11, équation du temps du 21/06 : il faut lire + 0,40 au lieu de + 40.

POST-SCRIPTUM

Je remercie Monsieur Robert SAGOT pour sa longue critique parfois constructive de la première partie de l'article. Si la majorité des points soulevés relèvent du purisme qui sied au spécialiste ; deux ou trois autres méritent un rectificatif :

- Ce qui est appelé **point de l'écliptique** sur le dessin du cadran et dans le texte, est en fait le **centre gnomonique** du cadran.

- La sous-style est la projection du style polaire sur le cadran. Dans le cas décrit ici, le style doit donc être appelé plus précisément **style droit**.

- **Analemme** : Celui-ci est défini comme la projection orthographique

des cercles de la sphère céleste sur le plan du méridien. Il convient donc de remplacer le mot analemme par **courbe en forme de 8**, ou **indicatrice de temps légal**. En conséquence de quoi, et pour apaiser ceux que l'extension de la terminologie cadranique inquiète ; l'adjectif analemme ne doit pas qualifier la finalité du cadran (correction pour la latitude, la longitude, la déclinaison du Soleil et l'équation du temps) mais l'origine géométrique de son calcul ! Une impérieuse nécessité s'impose donc : trouver un autre nom au type de cadran décrit dans cet article ! Le concours est ouvert aux latinistes, hélienistes et autres étymologistes ... distingués ou non. Pour exemple, et par souci d'apporter ma modeste contribution à cette entreprise nationale, je propose : **CADRAN A STREPTOGRAMME HOROPHORE**. Si vous trouvez plus compliqué ... ou plus simple ... n'hésitez pas !

PROGRAMME DE CALCUL D'UN CADRAN SOLAIRE ANALEMMATIQUE

Suite à l'article de Michel DEROUET sur le cadran solaire analemmatique, voici pour le compléter, un programme, en basic standard presque indispensable...

Pour la clarté de celui-ci, je me suis efforcé de suivre la chronologie de l'article, de garder le même nom des variables des équations dans les calculs.

Attention, certains ordinateurs sont en mode Radians dès la mise sous tension. Pour une bonne compré-

hension et pallier à cela, ligne 240 le mode Degrés est demandé pour la suite des calculs.

Lignes :

270 à 640 Entrée de la longitude et de la latitude du lieu, conversion en degrés décimaux puis affichage.

660 à 690 Entrée de la hauteur du style.

710 à 770 Calcul de PE et DE, soient les distances point d'écliptique au style, et, du style à la droite d'équinoxe.

780 à 840 Affichage des calculs précédents.

870 à 990 Sortie des mesures servant à tracer les lignes des heures.

1020 à 1050 Entrée du nombre de dates retenues.

1090 à 1190 Entrée de la déclinaison du Soleil et de l'équation du temps pour chaque date retenue.

1210 à 1620 Calcul de la longueur de l'ombre sur chaque ligne horaire pour chaque 1/2 heure de chaque date, pour tracer les arcs diurnes et, affichage.

1640 à 1830 Calcul de la hauteur du Soleil pour chaque heure de chaque date.

1850 à 2260 Sortie de tous les données et calculs ; seuls les résultats des lignes 2190 à 2200 servent à tracer le cadran. Soit 1 - Une distance depuis la base du style sur la ligne de midi, vers le Nord si positif, vers le Sud si négatif. 2 - Une distance depuis ce point, perpendiculairement, vers l'Ouest si négatif, vers l'Est si positif.

2290 à 2420 Soit sous-programme de calcul.

Daniel DENEUCHATEL

A VOS CLAVIERS...

```

100 REM *****
110 REM * PROGRAMME EN BASIC STANDARD *
120 REM * POUR LE CALCUL D'UN *
130 REM * CADRAN SOLAIRE ANALEMMATIQUE *
140 REM * PAR Daniel DENEUCHATEL *
150 REM * D'APRES L'ARTICLE DE *
160 REM * Michel DEROUET *
170 REM *****
180 REM
190 REM -INITIALISATION DES VARIABLES-
200 REM
210 DIM DM(11),DATE$(30),MM(30,12)
220 DIM D(30),EQ(30),M(30,12),HS(30,12),
A1(30,12),A2(30,12),A3(30,12)
230 REM
240 REM --MODE DEGRES-EFFACE ECRAN--
250 REM
255 DEG:CLS
260 REM
270 REM -ENTRER LES COORDONNEES DU LIEU-
280 REM - LATITUDE
290 REM
300 PRINT "ENTRER LA LATITUDE ET LA LONGI-
TUDE"
310 PRINT "DU LIEU EN (Deg,[R],mn,[R],se-
c,[R]):";PRINT "[R] = TOUCHE RETOUR"
320 PRINT:INPUT "LATITUDE Deg =";D1
330 INPUT " mn =";M1
340 INPUT " sec =";S1
350 REM
360 REM -CONVERSION EN DEGRES DECIMAUX-
370 REM
380 LA=SGN(D1)*(ABS(D1)+M1/60+S1/3600)
390 REM
400 REM -EST OU OUEST DE GREENWICH-
410 REM
420 CLS:PRINT "VOTRE LIEU EST-IL A L'EST
DE GREENWICH"

```

```

430 PRINT "TAPER [E] POUR L'EST ET [O] P-
OUR L'OUEST"
440 INPUT " O/E =";G$
450 IF G$="O"OR G$="E" THEN GOTO 490 ELS-
E GOTO 420
460 REM -----
470 REM - LONGITUDE
480 REM -----
490 INPUT "LONGITUDE Deg =";G
500 IF G$="O" THEN A=-1 ELSE A=1
510 IF G$="O"AND G=0 THEN A$=" " ELSE A$-
="":G=ABS(G)
520 INPUT " mn =";M2
530 INPUT " sec =";S2
540 REM -----
550 REM -CONVERSION EN DEGRES DECIMAUX-
560 REM -----
570 LO=A*(ABS(G)+M2/60+S2/3600)
580 REM -----
590 REM -AFFICHAGE LA ET LO-
600 REM -----
610 CLS:PRINT "LATITUDE =";D1;"Deg";M1;
";S1;"
620 PRINT "LONGITUDE =";A$;G;"Deg";M2;";
";S2;"
630 PRINT
640 PRINT:PRINT "TAPER RETOUR POUR CONTI-
NUER ";INPUT "R$";
650 REM -----
660 REM -ENTRER LA HAUTEUR DU STYLE-
670 REM -----
680 CLS:PRINT "DONNER LA HAUTEUR DU STYL-
E"
690 INPUT "EN MILLIMETRES =";ST:CLS
700 REM -----
710 REM -CALCUL DE PE-
720 REM -----

```



```

730 PE=ST*(1/TAN(LA))
740 REM -----
750 REM -CALCUL DE DE-
760 REM -----
770 DE=ST*TAN(LA)
780 PRINT "HAUTEUR DU STYLE EST EGAL A "
790 PRINT "mm";PRINT
790 PRINT "LA DISTANCE DU POINT D'ECLIPSE
IQUE AU "
800 PRINT "STYLE EST EGAL A ";PE;" mm"
810 PRINT
820 PRINT "LA DISTANCE DE LA DROITE D'EQ
UINOXES AU "
830 PRINT "STYLE EST EGAL A ";DE;" mm"
840 PRINT;PRINT "TAPEZ RETOUR POUR CONTI
NUER ";INPUT " ";R$
850 CLS
860 REM -----
870 REM -SORTIE DES MESURES SUR-
875 REM - LA LIGNE D'EQUINOXE -
880 REM -----
890 PRINT "LES MESURES SUIVANTES SONT A
REPORTER DE"
900 PRINT "PART ET D'AUTRE DE LA LIGNE D
U MIDI."
910 PRINT "TRACER LES DROITES DES HEURE
S PARTANT "
920 PRINT "DU POINT D'ECLIPSE EN PASS
ANT PAR CES"
930 PRINT "REPÈRES."
940 FOR I=1 TO 11
950 H=7.5*I
960 DM(I)=TAN(H)*ST/COS(LA)
970 PRINT USING "MESURE No ##      ";I;PRIN
T USING "###.###";DM(I)
980 NEXT I
990 PRINT;PRINT "TAPEZ RETOUR POUR CONTI
NUER ";INPUT " ";R$
1000 CLS
1010 REM -----
1020 REM -NOMBRE DE DATES-
1030 REM -----
1040 PRINT "ENTRER LE NOMBRE DE DATES UT
ILISEES "
1050 INPUT "      " =ND
1060 REM -----
1070 REM -DECLINAISON,EQUATION DU TEMPS-
1080 REM -----
1090 CLS;PRINT "      ENTRER POUR LA D
ATE DONNEE LA "
1100 PRINT "DECLINAISON ET L'EQUATION
DU TEMPS EN "
1110 PRINT "DEGRES DECIMAUX."
1120 FOR J=1 TO ND
1130 PRINT
1140 PRINT USING "      DATE No ##      JJ/MM
";J;INPUT " ";DATE$(J)
1150 PRINT
1160 INPUT "DECLINAISON DU SOLEIL DD.dd:
";D(J)
1170 PRINT
1180 INPUT "L'EQUATION DU TEMPS      DD.dd:
";EQ(J)
1190 NEXT J:CLS
1200 REM -----
1210 REM - CALCUL DES LONGUEURS DES -
1220 REM - OMBRES POUR TRACER LES ARCS -
1230 REM - DIURNES -
1240 REM -----
1250 FOR J=1 TO ND:K=0
1260 PRINT USING "LE ##### ";DATE$(J);
1270 FOR H=0 TO 90 STEP 7.5
1280 ZZ=ASN(SIN(LA)*SIN(D(J))+COS(LA)*CO
S(D(J))*COS(H))
1290 IF ZZ=0 OR ZZ<0 THEN MM(J,K)=0:PRIN
T ". ";GOTO 1320
1300 MM(J,K)=ST/TAN(ZZ)
1310 PRINT " ";
1320 K=K+1
1330 NEXT H
1340 PRINT;NEXT J
1350 REM -----
1360 REM - AFFICHAGE DES MESURES PAR -
1370 REM - DATES ET PAR HEURES -
1380 REM -----
1390 K=0:CLS
1400 PRINT "-----"
1410 PRINT "DONNEES POUR TRACER LES AR
CS DIURNES "
1420 PRINT "CES MESURES SONT A REPPORT
ER SUR LES "
1430 PRINT "LIGNES DES HEURES DE PART
ET D'AUTRE "
1440 PRINT "DE CELLE DE MIDI."
1450 PRINT "-----"
1460 PRINT;PRINT "TAPEZ RETOUR POUR CONT
INUER ";INPUT " ";R$
1470 CLS
1480 FOR J=1 TO ND
1490 I=12:PRINT "-----"
1500 PRINT USING "MESURES POUR LE #####
";DATE$(J)
1510 PRINT "-----"
1520 PRINT USING "##### H
";I;MM(J,K)
1530 FOR I=12.5 TO 18 STEP .5
1540 K=K+1
1550 IF MM(J,K)=0 THEN GOTO 1560 ELSE GO
TO 1580
1560 PRINT USING "##### H et ##### H s
ous l'horizon ";I;12-(I-12)

```

```

1570 GOTO 1590
1580 PRINT USING "##### H et ##### H
";I;12-(I-12);MM(J,K)
1590 NEXT I
1600 PRINT "-----"
1610 PRINT "TAPEZ RETOUR POUR CONTINUER
";INPUT " ";R$
1620 CLS;K=0:NEXT J
1630 REM -----
1640 REM - HAUTEUR DU SOLEIL AU -
1650 REM - DESSUS DE L'HORIZON POUR -
1660 REM - CHAQUE HEURE DE CHAQUE DATE-
1670 REM -----
1680 K=0:CLS
1690 PRINT "CALCULS ET REMPLISSAGE DES
TABLEAUX "
1700 PRINT
1710 FOR J=1 TO ND
1720 PRINT USING "LE ##### ";DATE$(J)
1730 FOR I=75 TO 15 STEP -15
1740 H2=-I+L0-EQ(J);K=K+1
1750 GOSUB 2360
1760 PRINT " ";NEXT I
1770 H2=0+L0-EQ(J);K=K+1;GOSUB 2360;PRIN
T " "
1780 FOR I=15 TO 75 STEP 15
1790 H2=I+L0-EQ(J);K=K+1
1800 GOSUB 2360
1810 PRINT " ";NEXT I
1820 K=0
1830 PRINT;NEXT J
1840 REM -----
1850 REM - AFFICHAGE PAR DATE DE : -
1860 REM - LA DECLINAISON DU SOLEIL -
1870 REM - L'EQUATION DU TEMPS -
1880 REM - L'HEURE -
1890 REM - LA HAUTEUR DU SOLEIL -
1900 REM - LA LONGUEUR DE L'OMBRE -
1910 REM - L'ANGLE A1 -
1920 REM - L'ANGLE A2 -
1930 REM - L'ANGLE A3 -
1940 REM - LE 1er REPPORT -
1950 REM - LE 2eme REPPORT -
1960 REM -----
1970 CLS
1980 PRINT "-----"
1990 PRINT "DONNEES POUR TRACER LES
ANALEMMES "
2000 PRINT "POUR CHAQUE HEURE DE CHA
QUE DATE "
2010 PRINT "-----"
2020 PRINT;PRINT "TAPEZ RETOUR POUR CONT
INUER ";INPUT " ";R$
2030 CLS;K=0
2040 FOR I=7 TO 17
2050 K=K+1
2060 FOR J=1 TO ND
2070 PRINT "-----"
2080 PRINT USING "MESURE POUR LE #####
";DATE$(J);I
2090 PRINT "-----"
2100 PRINT USING "DECLINAISON DU SOLEIL
";D(J)
2110 PRINT USING "EQUATION DU TEMPS
";EQ(J)
2120 IF HS(J,K)=0 THEN PRINT "SOLEIL SOUS L'HORIZON "
2130 IF HS(J,K)=0 THEN 2210
2140 PRINT USING "HAUTEUR DU SOLEIL
";HS(J,K)
2150 PRINT USING "LONGUEUR DE L'OMBRE
";M(J,K)
2160 PRINT USING "ANGLE A1
";A1(J,K)
2170 PRINT USING "ANGLE A2
";A2(J,K)
2180 PRINT USING "ANGLE A3
";A3(J,K)
2190 PRINT USING "1 er REP. DISTANCE AU
STYLE:#####";COS(A3(J,K))*M(J,K)
2200 PRINT USING "2 eme REP. DISTANCE A
U MIDI:#####";SIN(A3(J,K))*M(J,K)
2210 PRINT "-----"
2220 PRINT
2230 PRINT;PRINT "TAPEZ RETOUR POUR CONT
INUER ";INPUT " ";R$
2240 CLS
2250 NEXT J
2260 NEXT I
2270 END
2280 REM *****
2290 REM *SOUS PROGRAMME DE CALCUL *
2300 REM *HAUTEUR DU SOLEIL *
2310 REM *LONGUEUR DE L'OMBRE *
2320 REM *ANGLE A1 *
2330 REM *ANGLE A2 *
2340 REM *ANGLE A3 *
2350 REM *****
2360 HS(J,K)=ASN(SIN(LA)*SIN(D(J))+COS(L
A)*COS(D(J))*COS(H2))
2370 IF HS(J,K)<=0 THEN HS(J,K)=0:GOTO 2
420
2380 M(J,K)=ST/TAN(HS(J,K))
2390 A1(J,K)=ATN(TAN(H2)*(ST/COS(LA))/(D
E+PE))
2400 A2(J,K)=ASN((SIN(A1(J,K))*PE)/M(J,K
))
2410 A3(J,K)=A1(J,K)+A2(J,K)
2420 RETURN

```